
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 7.2.3.1, which is part of the LTE test suite. The TTCN changes done for test case 8.1.2.1 submitted in CR R5s090137 are applicable to this test case.

The test case can be demonstrated to run with one LTE UE (see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
tc_7_2_3_1
Test Group:
RRC

ATS Version:
iwd-EUTRA-TVB2009-03_D09wk38
System Simulator used:
Anite CT
UE used:
LG Electronics LEO3 UE.

Verification Status:
PASS
4 Corrections required for test case 7.2.3.1

4.1 Introduction

This section describes the changes required to make test case 7.2.3.1 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-TVB2009-03_D09wk38 release.

4.2 Change 1

Testcase name
f_RxAMD_PDU

Reason for change
When the RLC PDU is received with poll bit set, then SS needs to send the ACK.

Summary of change
Changed to send the ACK when the RLC PDU is received with poll bit set.

Source of change
File : RLC_AM_Testcases.ttcn

Before:

function f_RxAMD_PDU (inout RLC_SS_State p_RLC_Rec,

                      B1_Type p_Poll,

                      RLC_FramingInfo_Type p_FramingInfo,

                      template RLC_PDU_Header_FlexPart_Type p_Header_FlexPart,

                      RLC_DataFieldList_Type p_Data

)

runs on EUTRA_PTC

{

  DRB.receive (car_RLCAmDataInd (cr_AMD_PDU_LIs (p_RLC_Rec.AM_VRR,

                                                 p_Poll,

                                                 p_FramingInfo,

                                                 p_Header_FlexPart,

                                                 p_Data)));

  p_RLC_Rec.AM_VRR := (p_RLC_Rec.AM_VRR + 1) mod 1024;

} // f_RxAMD_PDU

After:

function f_RxAMD_PDU (inout RLC_SS_State p_RLC_Rec,

                      B1_Type p_Poll,

                      RLC_FramingInfo_Type p_FramingInfo,

                      template RLC_PDU_Header_FlexPart_Type p_Header_FlexPart,

                      RLC_DataFieldList_Type p_Data

)

runs on EUTRA_PTC

{

  DRB.receive (car_RLCAmDataInd (cr_AMD_PDU_LIs (p_RLC_Rec.AM_VRR,

                                                 p_Poll,

                                                 p_FramingInfo,

                                                 p_Header_FlexPart,

                                                 p_Data)));

  p_RLC_Rec.AM_VRR := (p_RLC_Rec.AM_VRR + 1) mod 1024;

  if ( p_Poll == '1'B)

  {

    f_TxSTATUS_PDU(p_RLC_Rec);         

  }
} // f_RxAMD_PDU

4.3 Change 2

Testcase name
fl_EUTRA_7_2_3_1

Reason for change
As per 36.331 RadioResourceConfigDedicated definition IE “eps-BearerIdentity” , “locicalChannelIdentity “and “pdcp-Config” should only be present in case of setup of a DRB or during handover. Hence in this case these IE should be set to OMIT. 


Summary of change
Changed the template set the parameters to OMIT.

Source of change
File : EUTRA_RRC_Templates.ttcn

Before:

template (value) RadioResourceConfigDedicated cds_SRB2_1AM_DRB_Config_RLC_modif ( template (omit) CQI_ReportPeriodic p_CQI_ReportingPeriodic,

                                                                                  template (omit) SoundingRS_UL_ConfigDedicated p_SoundingRsUl_ConfigDedicated,

                                                                                  template (value) UL_AM_RLC p_UL_AM_RLC,

                                                                                  template (value) DL_AM_RLC p_DL_AM_RLC)

  modifies cs_SRB2_1AM_DRB_Config_Def := {

  drb_ToAddModList := {

    {

      rlc_Config := {

        am := {

          ul_AM_RLC := p_UL_AM_RLC,

          dl_AM_RLC := p_DL_AM_RLC

        }

      }

    }

  }

};

After:

template (value) RadioResourceConfigDedicated cds_SRB2_1AM_DRB_Config_RLC_modif ( template (omit) CQI_ReportPeriodic p_CQI_ReportingPeriodic,

                                                                                  template (omit) SoundingRS_UL_ConfigDedicated p_SoundingRsUl_ConfigDedicated,

                                                                                  template (value) UL_AM_RLC p_UL_AM_RLC,

                                                                                  template (value) DL_AM_RLC p_DL_AM_RLC)

  modifies cs_SRB2_1AM_DRB_Config_Def_RLC:= {

  drb_ToAddModList := {

    {

      rlc_Config := {

        am := {

          ul_AM_RLC := p_UL_AM_RLC,

          dl_AM_RLC := p_DL_AM_RLC

        }

      }

    }

  }

};

  template (value) RadioResourceConfigDedicated cs_SRB2_1AM_DRB_Config_Def_RLC ( template (omit) CQI_ReportPeriodic p_CQI_ReportingPeriodic,

                                                                             template (omit) SoundingRS_UL_ConfigDedicated p_SoundingRsUl_ConfigDedicated ) :=

  /* Values according to 36.508 cl. 4.6.3 Table 4.6.3-16 (using condition n=1 & m=0) */

    cs_508_RadioResourceConfigDedicated_Srb2( { cs_508_DRB_ToAddMod_DEFAULT_AM_RLC(tsc_DRB1) },

                                              cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                              cs_508_PhysicalConfigDedicated_Default_RBC ( p_CQI_ReportingPeriodic, p_SoundingRsUl_ConfigDedicated ) );



  template (value) DRB_ToAddMod cs_508_DRB_ToAddMod_DEFAULT_AM_RLC (DRB_Identity p_DRB_Id) :=

  // According to 36.508 cl. 4.8.2.1.7 Table 4.8.2.1.7-1 (using condition AM)

    cs_508_DRB_ToAddMod_Common_RLC(p_DRB_Id,

                               cs_508_PDCP_Config_DRB_AM,

                               cs_508_RLC_Config_DRB_AM,

                               cs_508_LogicalChannelConfig_DRB_AM);     

///Anite_LTE_114

  template (value) DRB_ToAddMod cs_508_DRB_ToAddMod_Common_RLC(DRB_Identity p_DRB_Id,

                                                           template (omit) PDCP_Config  p_PDCP_Config,

                                                           template (omit) RLC_Config   p_RLC_Config,

                                                           template (omit) LogicalChannelConfig p_LogicalChannelConfig) :=

  { // common template which deals with assignment of eps_BearerIdentity and logicalChannelIdentity

    eps_BearerIdentity := omit, //p_DRB_Id+4,

    drb_Identity := p_DRB_Id,

    pdcp_Config := omit, //p_PDCP_Config

    rlc_Config := p_RLC_Config,

    logicalChannelIdentity := omit, // p_DRB_Id+2,

    logicalChannelConfig := p_LogicalChannelConfig

  };

4.4 Change 3

Testcase name
fl_EUTRA_7_2_3_1

Reason for change
There is no default grant transmission for the scheduling request received after the test body.

Summary of change
Enabled the default grants after the test body. 

Source of change
File : RLC_AM_Testcases.ttcn

Before:

  f_EUTRA_TestBody_Set(true);

  fl_EUTRA_7_2_3_1_TestBody (v_RLC_Rec);

  f_EUTRA_TestBody_Set(false);

  // leaving UE test state 4

  f_OpenUE_TestLoopMode_Deactivate_TestMode ( eutra_Cell1 );

  // postamble

  f_EUTRA_Postamble (eutra_Cell1, RRC_CONNECTED, tsc_RRC_TI_Def, ATTACHED);

After:

  f_EUTRA_TestBody_Set(true);

  fl_EUTRA_7_2_3_1_TestBody (v_RLC_Rec);

  f_EUTRA_TestBody_Set(false);

  f_EUTRA_StartDefULGrantTransmission(eutra_Cell1,cs_TimingInfo_Now);
  // leaving UE test state 4

  f_OpenUE_TestLoopMode_Deactivate_TestMode ( eutra_Cell1 );

  // postamble

  f_EUTRA_Postamble (eutra_Cell1, RRC_CONNECTED, tsc_RRC_TI_Def, ATTACHED);

4.5 Change 4

Testcase name
fl_EUTRA_7_2_3_1_TestBody

Reason for change
1) At Step 1 SS is configured to disable SR reporting. This is not required as by default SS is always configured in a mode not to report SR.

2) After receiving the scheduling request, further Scheduling request indication to TTCN should be disabled and grants should be allocated for 2 RLC PDU. 

3) RLC needs to wait for finite amount of time to receive the status PDU.

4) When sending and receiving the 3 RLC PDU the data that needs to be taken is from RLC data list 3, 4 and 5. 

Summary of change
1) Disabled the scheduling request indication disabling. 

2) Changed to stop the scheduling request indication to TTCN. Changed the number of grants to be sent to 2 repetitions.

3) Added the timer to receive the status PDU.

4) Changed the Data list to 3 4 and 5.

Source of change
File : RLC_AM_Testcases.ttcn

Before:

function fl_EUTRA_7_2_3_1_TestBody(inout RLC_SS_State p_RLC_Rec)

runs on EUTRA_PTC {

  var integer i;

  var RLC_DataField_Type v_RLC_Data1, v_RLC_Data2, v_RLC_Data3;

  var template (value) PDCP_DataPdu_LongSN_Type v_PDCP_DataPDU;

  // timer value to be used as per configuration

  timer t_WatchDog := 5.0;

  // PDCP SDU size employed: 38 bytes

  var integer v_PDCP_SDU_size := 38;

  //PDCP SDUs of this size will be built, appended to a PDCP header,

  //and then used as RLC SDUs

  // generate 5 RLC SDUs = PDCP PDUs

  for(i := 0; i < 5; i:= i + 1) {

    p_RLC_Rec.TxDataSize := v_PDCP_SDU_size;

    f_GetTxAM_PRBS (p_RLC_Rec);  // sets p_RLC_Rec.RLC_Data

    v_PDCP_DataPDU := cs_PDCP_DataPdu_LongSN2(int2bit(i, tsc_PDCP_SN_Size_12 ), p_RLC_Rec.RLC_Data);

    p_RLC_Rec.RLC_DataList[i] := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(v_PDCP_DataPDU));

    p_RLC_Rec.Next_PDCP_TX_SN := (p_RLC_Rec.Next_PDCP_TX_SN + 1) mod tsc_Maximum_PDCP_SN_12;

  }

  // Step 1: The SS ignores scheduling requests and does not allocate any uplink grant

  f_SS_ConfigSchedulingRequestIndMode(eutra_Cell1, tsc_L1Mac_IndicationMode_Disable); //NOTE1
  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 2: SS transmits an AMD PDU including two RLC SDUs of size 40 bytes each with poll bit set to '1'.

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 0, 40);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 0, 40);

  f_TxAMD_PDU_List( p_RLC_Rec,

                    tsc_P_Poll,

                    tsc_FI_FullSDU,

                    {v_RLC_Data1, v_RLC_Data2});

  // Step 3: SS responds to any scheduling requests from the UE by transmitting UL grants of size 776 bits (Note 1).

  f_SS_ConfigSchedulingRequestIndMode(eutra_Cell1, tsc_L1Mac_IndicationMode_Enable);

  // Wait for SR

  t_WatchDog.start;

  alt {

    [] SYSIND.receive ( car_SchedulingReq_IND (eutra_Cell1, ?))

     {

       t_WatchDog.stop;

     }

    [] t_WatchDog.timeout

     {

       f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE is not sending scheduling Request during the watchdog period");

     }

  }; // alt

  f_EUTRA_PeriodicCyclicULGrantTransmission( eutra_Cell1, cs_TimingInfo_Now, 14, 3, 1, 1); // Imcs = 14 -> Itbs = 13 //NOTE2
  // Step 4: Check: Does UE transmit a STATUS PDU with positive acknowledgement ?

  p_RLC_Rec.AM_VTA := 1; // needed or not ?

  f_RxSTATUS_PDU_ACK ( p_RLC_Rec );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");

  // Step 5: Check: Does UE transmit two RLC SDUs within an AMD PDU with FI field set to '00',

  // first E field in the fixed part set to '1', first E field in the extension part set to '0', first LI field set to 40 bytes?

  // concatenate both SDUs

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 0, 40);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 0, 40);

  f_RxAMD_PDU (p_RLC_Rec,

               tsc_P_Poll,

               tsc_FI_FullSDU,

               cr_FlexPart_1LI (40),

               {v_RLC_Data1, v_RLC_Data2} );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");

  // Step 6: The SS ignores scheduling requests and does not allocate any uplink grant.

  f_SS_ConfigSchedulingRequestIndMode(eutra_Cell1, tsc_L1Mac_IndicationMode_Disable);
  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 7: SS transmits an AMD PDU including three RLC SDU of size 40 bytes with P field set to "1".

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 0, 40);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 0, 40);

  v_RLC_Data3 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 0, 40);

  f_TxAMD_PDU_List ( p_RLC_Rec,

                     tsc_P_Poll,

                     tsc_FI_FullSDU,

                     {v_RLC_Data1, v_RLC_Data2, v_RLC_Data3} );

  // Step 8: SS responds to any scheduling requests from the UE by transmitting an UL grant of size 1096 bits. (Note 2)

  f_SS_ConfigSchedulingRequestIndMode(eutra_Cell1, tsc_L1Mac_IndicationMode_Enable);

  // Wait for SR

  t_WatchDog.start;

  alt {

    [] SYSIND.receive ( car_SchedulingReq_IND (eutra_Cell1, ?))

     {

       t_WatchDog.stop;

     }

    [] t_WatchDog.timeout

     {

       f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE is not sending scheduling Request during the watchdog period");

     }

  }; // alt

  f_EUTRA_PeriodicCyclicULGrantTransmission( eutra_Cell1, cs_TimingInfo_Now, 8, 8, 1, 1); // Imcs = 8 -> Itbs = 8
  // Step 9: Check: Does UE transmits a STATUS PDU with positive acknowledgement?

  p_RLC_Rec.AM_VTA := 2; // needed or not ?

  f_RxSTATUS_PDU_ACK ( p_RLC_Rec );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 9");

  // Step 10: Check: Does UE transmit three RLC SDUs within an AMD PDU with FI field set to "00", first E field in the fixed part set to '1',

  // first E field in the extension part set to '1', first LI field set to 40 bytes, second E field in the extension part set to '0',

  // second LI field set to 40 bytes and P field set to "1"?

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 0, 40);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 0, 40);

  v_RLC_Data3 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 0, 40);

  f_RxAMD_PDU (p_RLC_Rec,

               tsc_P_Poll,

               tsc_FI_FullSDU,

               cr_FlexPart_2LIs (40, 40),

               {v_RLC_Data1, v_RLC_Data2, v_RLC_Data3} );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 10");

} // fl_EUTRA_7_2_3_1_TestBody

After: 

 function fl_EUTRA_7_2_3_1_TestBody(inout RLC_SS_State p_RLC_Rec)

runs on EUTRA_PTC {

  var integer i;

  var RLC_DataField_Type v_RLC_Data1, v_RLC_Data2, v_RLC_Data3;

  var template (value) PDCP_DataPdu_LongSN_Type v_PDCP_DataPDU;

  // timer value to be used as per configuration

  timer t_WatchDog := 5.0;

  // PDCP SDU size employed: 38 bytes

  var integer v_PDCP_SDU_size := 38;

  //PDCP SDUs of this size will be built, appended to a PDCP header,

  //and then used as RLC SDUs

  // generate 5 RLC SDUs = PDCP PDUs

  for(i := 0; i < 5; i:= i + 1) {

    p_RLC_Rec.TxDataSize := v_PDCP_SDU_size;

    f_GetTxAM_PRBS (p_RLC_Rec);  // sets p_RLC_Rec.RLC_Data

    v_PDCP_DataPDU := cs_PDCP_DataPdu_LongSN2(int2bit(i, tsc_PDCP_SN_Size_12 ), p_RLC_Rec.RLC_Data);

    p_RLC_Rec.RLC_DataList[i] := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(v_PDCP_DataPDU));

    p_RLC_Rec.Next_PDCP_TX_SN := (p_RLC_Rec.Next_PDCP_TX_SN + 1) mod tsc_Maximum_PDCP_SN_12;

  }

  // Step 1: The SS ignores scheduling requests and does not allocate any uplink grant

//  f_SS_ConfigSchedulingRequestIndMode(eutra_Cell1, tsc_L1Mac_IndicationMode_Disable);   
  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 2: SS transmits an AMD PDU including two RLC SDUs of size 40 bytes each with poll bit set to '1'.

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 0, 40);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 0, 40);

  f_TxAMD_PDU_List( p_RLC_Rec,

                    tsc_P_Poll,

                    tsc_FI_FullSDU,

                    {v_RLC_Data1, v_RLC_Data2});

  // Step 3: SS responds to any scheduling requests from the UE by transmitting UL grants of size 776 bits (Note 1).

  f_SS_ConfigSchedulingRequestIndMode(eutra_Cell1, tsc_L1Mac_IndicationMode_Enable);

  // Wait for SR

  t_WatchDog.start;

  alt {

    [] SYSIND.receive ( car_SchedulingReq_IND (eutra_Cell1, ?))

     {

       t_WatchDog.stop;

     }

    [] t_WatchDog.timeout

     {

       f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE is not sending scheduling Request during the watchdog period");

     }

  }; // alt

  f_SS_ConfigSchedulingRequestIndMode(eutra_Cell1, tsc_L1Mac_IndicationMode_Disable);   

  f_EUTRA_PeriodicCyclicULGrantTransmission( eutra_Cell1, cs_TimingInfo_Now, 14, 3, 1, 2); // Imcs = 14 -> Itbs = 13 //NOTE2
  // Step 4: Check: Does UE transmit a STATUS PDU with positive acknowledgement ?

  p_RLC_Rec.AM_VTA := 1; // needed or not ?

  f_RxSTATUS_PDU_ACK ( p_RLC_Rec );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");

  // Step 5: Check: Does UE transmit two RLC SDUs within an AMD PDU with FI field set to '00',

  // first E field in the fixed part set to '1', first E field in the extension part set to '0', first LI field set to 40 bytes?

  // concatenate both SDUs

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 0, 40);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 0, 40);

  f_RxAMD_PDU (p_RLC_Rec,

               tsc_P_Poll,

               tsc_FI_FullSDU,

               cr_FlexPart_1LI (40),

               {v_RLC_Data1, v_RLC_Data2} );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");

  t_WatchDog.start(500E-3);

  t_WatchDog.timeout; //NOTE3
  // Step 6: The SS ignores scheduling requests and does not allocate any uplink grant.

//  f_SS_ConfigSchedulingRequestIndMode(eutra_Cell1, tsc_L1Mac_IndicationMode_Disable);   //NOTE1
  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 7: SS transmits an AMD PDU including three RLC SDU of size 40 bytes with P field set to "1".

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 0, 40);  //NOTE4
  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4], 0, 40);  //NOTE4
  v_RLC_Data3 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 0, 40);  //NOTE4
  f_TxAMD_PDU_List ( p_RLC_Rec,

                     tsc_P_Poll,

                     tsc_FI_FullSDU,

                     {v_RLC_Data1, v_RLC_Data2, v_RLC_Data3} );

  // Step 8: SS responds to any scheduling requests from the UE by transmitting an UL grant of size 1096 bits. (Note 2)

  f_SS_ConfigSchedulingRequestIndMode(eutra_Cell1, tsc_L1Mac_IndicationMode_Enable);

  // Wait for SR

  t_WatchDog.start;

  alt {

    [] SYSIND.receive ( car_SchedulingReq_IND (eutra_Cell1, ?))

     {

       t_WatchDog.stop;

     }

    [] t_WatchDog.timeout

     {

       f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE is not sending scheduling Request during the watchdog period");

     }

  }; // alt

  f_SS_ConfigSchedulingRequestIndMode(eutra_Cell1, tsc_L1Mac_IndicationMode_Disable);   

  f_EUTRA_PeriodicCyclicULGrantTransmission( eutra_Cell1, cs_TimingInfo_Now, 8, 8, 1, 2); // Imcs = 8 -> Itbs = 8 //NOTE2
  // Step 9: Check: Does UE transmits a STATUS PDU with positive acknowledgement?

  p_RLC_Rec.AM_VTA := 2; // needed or not ?

  f_RxSTATUS_PDU_ACK ( p_RLC_Rec );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 9");

  // Step 10: Check: Does UE transmit three RLC SDUs within an AMD PDU with FI field set to "00", first E field in the fixed part set to '1',

  // first E field in the extension part set to '1', first LI field set to 40 bytes, second E field in the extension part set to '0',

  // second LI field set to 40 bytes and P field set to "1"?

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 0, 40); //NOTE4
  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4], 0, 40); //NOTE4
  v_RLC_Data3 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 0, 40); //NOTE4
  f_RxAMD_PDU (p_RLC_Rec,

               tsc_P_Poll,

               tsc_FI_FullSDU,

               cr_FlexPart_2LIs (40, 40),

               {v_RLC_Data1, v_RLC_Data2, v_RLC_Data3} );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 10");

} // fl_EUTRA_7_2_3_1_TestBody

5 Execution Log Files

5.1 LG Electronics LEO3 UE

The LG electronics LEO3 UE passed this test case on the Anite LTE solution with Snow 3G integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file tc_7_2_3_1_LG-log.html:

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

6 References

[1]
R5s090161: This archive comprises text format execution log file and the TTCN file.

